

Gramática LM - LL1 EBNF

MODULE = (1) FUNCTIONS MAIN

FUNCTIONS = (2) 'ID' PARAMS RETURNTYPE ESCAPE FUNCTIONS | (3) ϵ

MAIN = (4) 'MAIN' 'OP' 'CP' 'VOID' ESCAPE

PARAMS = (5) 'OP' PARAMSEXT

PARAMSEXT = (6) 'CP' | (7) LISTPARAMS 'CP'

LISTPARAMS = (8) TYPE NAME LISTPARAMSEXT

LISTPARAMSEXT = (9) 'SPTR' LISTPARAMS | (10) ϵ

TYPE = (11) 'INT' | (12) 'BOOL' | (13) 'CHAR' | (14) 'STRING' | (15) 'FLOAT'

RETURNTYPE = (16) 'VOID' | (17) TYPE RETURNTYPEEXT

RETURNTYPEEXT = (18) 'OB' 'CB' | (19) ϵ

NAME = (20) 'ID' NAMEEXT

NAMEEXT = (21) 'OB' EA 'CB' | (22) ϵ

ESCAPE = (23) 'OK' COMMANDS 'CK' 'SCO'

COMMANDS = (24) CMD 'SCO' COMMANDS | (25) ϵ

CMD = (26) DECLARATION | (27) 'ID' CMDEXT | (28) PRINT | (29) READ | (30) IFELSE | (31) WHEN | (32) REPEATER | (33) RETURN

DECLARATION = (34) TYPE NAME

CMDEXT = (35) ATTRIBUTION | (36) FUNCTIONCALL

ATTRIBUTION = (37) NAMEEXT 'ATR' VALUE

VALUE = (38) ARRAY | (39) EA

ARRAY = (40) 'OB' ARRAYEXT

ARRAYEXT = (41) ELEMENTS 'CB' | (42) 'CB'

ELEMENTS = (43) CONSTANT ELEMENTSEXT

ELEMENTSEXT = (44) 'SPTR' ELEMENTS | (45) ϵ

FUNCTIONCALL = (46) 'OP' LISTPARAMSCALL 'CP'

LISTPARAMSCALL = (47) PARAMITEM LISTPARAMSCALLEXT | (48) ϵ

LISTPARAMSCALLEXT = (49) 'SPTR' LISTPARAMSCALL | (50) ϵ

PARAMITEM = (51) CONSTANT | (52) NAME

PRINT = (53) 'PRINT' 'OP' MESSAGE PRINTEXT 'CP'

PRINTEXT = (54) 'SPTR' NAME PRINTEXT | (55) ϵ

MESSAGE = (56) 'CTESTRING' MESSAGEEXT | (57) NAME MESSAGEEXT

MESSAGEEXT = (58) 'CONCAT' MESSAGE | (59) ϵ

READ = (60) 'READ' 'OP' READEXT 'CP'

READEXT = (61) NAME READEXTTR

READEXTTR = (62) 'SPTR' READEXT | (63) ϵ

IFELSE = (64) IF ELIF ELSE

IF = (65) 'IF' 'OP' EB 'CP' 'OK' COMMANDS 'CK'

ELIF = (66) 'ELIF' 'OP' EB 'CP' 'OK' COMMANDS 'CK' ELIF | (67) ϵ

ELSE = (68) 'ELSE' 'OK' COMMANDS 'CK' | (69) ϵ

WHEN = (70) 'WHEN' 'OP' EB 'CP' 'OK' COMMANDS 'CK'

REPEATER = (71) 'REPEATER' 'OP' 'ID' ATTRIBUTION 'SCO' PARAMITEM 'SCO'
PARAMITEM 'CP' 'OK' COMMANDS 'CK'

RETURN = (72) 'RETURN' RETURNEXT

RETURNEXT = (73) CONSTANT | (74) NAME

EB = (75) TB EBR

EBR = (76) 'OR' TB EBR | (77) ϵ

TB = (78) FB TBR

TBR = (79) 'AND' FB TBR | (80) ϵ

FB = (81) 'NOT' FB | (82) EA EREL

EREL = (83) 'ORC' EA | (84) 'ORE' EA

EA = (85) TA EAR

EAR = (86) 'OPA' TA EAR | (87) ϵ

TA = (88) PA TAR

TAR = (89) 'OPM' PA TAR | (90) ϵ

PA = (91) FA PAFAT

PAFAT = (92) 'OPE' PA | (93) ϵ

FA = (94) 'OPU' FA | (95) 'OP' EB 'CP' | (96) CONSTANT | (97) ID

ID = (98) 'ID' IDFAT

IDFAT = (99) 'OB' EA 'CB' | (100) 'OP' LISTPARAMSCALL 'CP' | (101) ϵ

CONSTANT = (102) 'CTEINT' | (103) 'CTEFLOAT' | (104) 'CTEBOOL' | (105) 'CTECHAR' |
(106) 'CTESTRING'