

Gramática LM - Sem Restrições

MODULE = (1) FUNCTIONS MAIN

FUNCTIONS = (2) 'ID' PARAMS RETURNTYPE ESCAPE FUNCTIONS | (3) ϵ

MAIN = (4) 'MAIN' 'OP' 'CP' 'VOID' ESCAPE

PARAMS = (5) 'OP' PARAMSEXT

PARAMSEXT = (6) 'CP' | (7) LISTPARAMS 'CP'

LISTPARAMS = (8) TYPE NAME LISTPARAMSEXT

LISTPARAMSEXT = (9) 'SPTR' LISTPARAMS | (10) ϵ

TYPE = (11) 'INT' | (12) 'BOOL' | (13) 'CHAR' | (14) 'STRING' | (15) 'FLOAT'

RETURNTYPE = (16) 'VOID' | (17) TYPE RETURNTYPEEXT

RETURNTYPEEXT = (18) 'OB' 'CB' | (19) ϵ

NAME = (20) 'ID' NAMEEXT

NAMEEXT = (21) 'OB' EA 'CB' | (22) ϵ

ESCAPE = (23) 'OK' COMMANDS 'CK' 'SCO'

COMMANDS = (24) CMD 'SCO' COMMANDS | (25) ϵ

CMD = (26) DECLARATION | (27) 'ID' CMDEXT | (28) PRINT | (29) READ | (30) IFELSE | (31) WHEN | (32) REPEATER | (33) RETURN

DECLARATION = (34) TYPE NAME

CMDEXT = (35) ATTRIBUTION | (36) FUNCTIONCALL

ATTRIBUTION = (37) NAMEEXT 'ATR' VALUE

VALUE = (38) ARRAY | (39) EA

ARRAY = (40) 'OB' ARRAYEXT

ARRAYEXT = (41) ELEMENTS 'CB' | (42) 'CB'

ELEMENTS = (43) CONSTANT ELEMENTSEXT

ELEMENTSEXT = (44) 'SPTR' ELEMENTS | (45) ϵ

FUNCTIONCALL = (46) 'OP' LISTPARAMSCALL 'CP'

LISTPARAMSCALL = (47) PARAMITEM LISTPARAMSCALLEXT | (48) ϵ

LISTPARAMSCALLEXT = (49) 'SPTR' LISTPARAMSCALL | (50) ϵ

PARAMITEM = (51) CONSTANT | (52) NAME

PRINT = (53) 'PRINT' 'OP' MESSAGE PRINTEXT 'CP'

PRINTEXT = (54) 'SPTR' NAME PRINTEXT | (55) ϵ

MESSAGE = (56) 'CTESTRING' MESSAGEEXT | (57) NAME MESSAGEEXT

MESSAGEEXT = (58) 'CONCAT' MESSAGE | (59) ϵ

READ = (60) 'READ' 'OP' READEXT 'CP'

READEXT = (61) NAME READEXTTR

READEXTTR = (62) 'SPTR' READEXT | (63) ϵ

IFELSE = (64) IF ELIF ELSE

IF = (65) 'IF' 'OP' EB 'CP' 'OK' COMMANDS 'CK'

ELIF = (66) 'ELIF' 'OP' EB 'CP' 'OK' COMMANDS 'CK' ELIF | (67) ϵ

ELSE = (68) 'ELSE' 'OK' COMMANDS 'CK' | (69) ϵ

WHEN = (70) 'WHEN' 'OP' EB 'CP' 'OK' COMMANDS 'CK'

REPEATER = (71) 'REPEATER' 'OP' 'ID' ATTRIBUTION 'SCO' PARAMITEM 'SCO'
PARAMITEM 'CP' 'OK' COMMANDS 'CK'

RETURN = (72) 'RETURN' RETURNEXT

RETURNEXT = (73) CONSTANT | (74) NAME

EB = (75) EB 'OR' TB | (76) TB

TB = (77) TB 'AND' FB | (78) FB

FB = (79) 'NOT' FB | (80) EA EREL

EREL = (81) 'ORC' EA | (82) 'ORE' EA

EA = (83) EA 'OPA' TA | (84) TA

TA = (85) TA 'OPM' PA | (86) PA

PA = (87) FA 'OPE' PA | (88) FA

FA = (89) 'OPU' FA | (90) 'OP' EB 'CP' | (91) CONSTANT | (92) ID

ID = (93) 'ID' | (94) 'ID' 'OB' EA 'CB' | (95) 'ID' 'OP' LISTPARAMSCALL 'CP'

CONSTANT = (96) 'CTEINT' | (97) 'CTEFLOAT' | (98) 'CTEBOOL' | (99) 'CTECHAR' | (100) 'CTESTRING'